

31

WYKORZYSTANIE ARKUSZA KALKULACYJNEGO DO WSPOMAGANIA PROCESU ROZLICZANIA KOSZTÓW UŻYTKOWANIA SAMOCHODÓW

31.1 WPROWADZENIE

Współczesne firmy zmuszone są do funkcjonowania i konkurowania ze sobą w niezwykle szybko zmieniającym się otoczeniu. Jednym z głównych czynników pozwalających na uzyskanie i utrzymanie wysokiego poziomu konkurencyjności w organizacji są jej dobrze zorganizowane i sprawnie funkcjonujące procesy. W tym zakresie firmy mogą wykorzystywać wiele różnych narzędzi informatycznych. Już w roku 2004 na rynku istniało kilkaset zróżnicowanych pod względem funkcjonalności rozwiązań, które mogły być zastosowane do wspomagania zarządzania [4], a liczba ta z pewnością cały czas rośnie. Niezależnie od wykorzystywanych w organizacji narzędzi informatycznych, nie są one jednak w stanie raz na zawsze rozwiązać wszystkich dynamicznie zmieniających się problemów w zakresie rejestrowania danych oraz ich przetwarzania. Wśród sytuacji, które mogą spowodować pojawienie się nowych wyzwań w tym zakresie można wyróżnić pojawienie się nowych nieprzewidzianych wcześniej problemów, konieczność spojrzenia na pewien wycinek przedsiębiorstwa w sposób bardziej szczegółowy niż miało to miejsce dotychczas, wprowadzenie nowych, innowacyjnych koncepcji w zakresie analizy danych [15]. Wprowadzenie modyfikacji w stosowanych systemach informatycznych wiąże się często z koniecznością zaangażowania pracowników firm zewnętrznych i przeznaczenia na ten cel pokaźnych środków pieniężnych. Wdrażanie zmian wymaga też znacznej ilości czasu, którym organizacje najczęściej nie dysponują. Dobrym rozwiązaniem możliwym do zastosowania w przypadku pojawienia się konieczności realizacji procesu informacyjnego, który nie jest objęty wykorzystywanymi w firmie rozwiązaniami informatycznymi jest szybkie stworzenie, przy wykorzystaniu posiadanych zasobów i umiejętności, własnych narzędzi, które mogą taki proces w znacznym stopniu zautomatyzować. Odpowiednim środowiskiem do tworzenia takich narzędzi mogą być Arkusze kalkulacyjne. Są one dostępne w prawie wszystkich organizacjach a zdecydowana większość pracowników deklaruje ich znajomość i wykorzystywanie przynajmniej w stopniu podstawowym. Grupa programów, których koncepcja pojawiła się po raz pierwszy w 1978 roku [8], a pierwszy przedstawiciel ujrzał światło dzienne w roku 1979 [21], ma za sobą okres bardzo intensywnego rozwoju. Od prostego narzędzia o nazwie VisiCalc w ciągu ponad 35 lat arkusze kalkulacyjne przeistoczyły się w potężne aplikacje

posiadające wiele bardziej lub mniej przydatnych elementów dających użytkownikom olbrzymie możliwości oraz znaczną elastyczność w zakresie gromadzenia i przetwarzania danych. Wraz z rozwojem znalazły one zastosowanie w wielu różnych obszarach przedsiębiorstwa. Problematyka arkuszy kalkulacyjnych jest obecnie poruszana w wielu artykułach i publikacjach książkowych. Dużo pozycji literaturowych omawia w szczególności sposób budowę oraz sposób wykorzystania ich funkcji i mechanizmów [6, 20]. W wielu publikacjach omawiane są również koncepcje wykorzystania arkuszy kalkulacyjnych do rozwiązania określonych problemów w różnych obszarach organizacji. Obejmują one wykorzystanie arkuszy kalkulacyjnych w obszarach zarządzania finansami, controllingu czy rachunkowości zarządczej [2, 3, 7, 22, 23], w obszarach związanych z zarządzaniem sprzedażą oraz marketingiem [2, 22, 23] czy w zakresie zarządzania jakością [2, 5, 14, 17, 18]. Ciekawą propozycją jest koncepcja wspomaganie przetwarzania danych z wykorzystaniem tworzonych w arkuszach kalkulacyjnych narzędzi bazujących na relacyjnym modelu danych. Przykładowe propozycje takich rozwiązań można znaleźć w pozycjach literaturowych [11, 12, 13, 14, 15, 16, 17, 18]. Szczegółowa koncepcja wykorzystania arkusza kalkulacyjnego, jako środowiska dla relacyjnej bazy danych została także zaprezentowana w publikacji [19]. Bazujące na relacyjnym modelu danych rozwiązanie wspomagające proces rozliczania kosztów użytkowania samochodów zostało zaprezentowane w dalszej części niniejszego opracowania.

31.2 OPIS PROCESU I ZAŁOŻONA FUNKCJONALNOŚĆ NARZĘDZIA

Firma, w której badany był proces rozliczania kosztów samochodów korzystała w zakresie zarządzania swoją flotą samochodową z usług firmy zewnętrznej. Podstawą do właściwego przyporządkowania kosztów związanych z użytkowaniem poszczególnych samochodów do odpowiednich miejsc powstawania kosztów, był raport przekazywaną przez firmę zewnętrzną. Raport ten był przesyłany, co miesiąc w postaci skoroszytu arkusza kalkulacyjnego MS Excel i zawierał zestawienie wszystkich zakupów towarów i usług dla każdego z posiadanych samochodów. Struktura otrzymywanego raportu została przedstawiona na rys. 31.1.

	A	B	C	D	E	F	G	H	I	J
	Rod Karty	Nr faktury	Nr karty	Nr rejestr.	Nazwa klienta	Data transakcji	Produkt	Ilość	Cena jed.	Wartość (bez VAT)
1										
2	BP	194445624	487	DW9876	Firma X	2015-05-11	Toll Service PL	1	14,00	14,00
3	BP	194445624	487	DW9876	Firma X	2015-05-14	SupraDiesel LTR	52	3,90	202,80

Rys. 31.1 Układ raportu otrzymywanego od firmy zewnętrznej

Źródło: opracowanie własne


W celu ustalenia kosztów, które mają zostać zaksięgowane obciążając poszczególne miejsca powstawania kosztów konieczne było wstępne przetworzenie posiadanych danych. Miejsce powstawania kosztów było ustalane na podstawie numeru rejestracyjnego pojazdu. Wysokość kosztu była uzależniona od rodzaju produktu oraz od faktu, czy

dany samochód był pojazdem wykorzystywanym wyłącznie do celów firmowych czy też był pojazdem wykorzystywanym zarówno do celów firmowych, jak i celów prywatnych pracowników. W przypadku niektórych produktów koszt obejmował wartość netto powiększoną o pełną wartość VAT. W przypadku pozostałych produktów możliwe były dwie sytuacje. W sytuacji, gdy samochód był używany wyłącznie do celów związanych z działalnością firmy, koszt obejmował jedynie wartość netto. W przypadku, gdy samochód wykorzystywany był zarówno do celów firmowych, jak i do celów prywatnych, koszt obejmował wartość netto powiększoną o 50% VAT. Stawka VAT mogła wynosić 8% lub 23%. W wyniku przetworzenia danych pochodzących z przesyłanego raportu, konieczne było uzyskanie informacji, którą można przedstawić w postaci raportu przedstawionego na rys. 31.2. Przetworzenie przez pracownika firmy danych zawartych w raporcie było operacją żmudną i pracochłonną. W związku z powyższym w celu usprawnienia procesu przetwarzania danych zaproponowane zostało narzędzie bazujące na relacyjnym modelu danych, stworzone w arkuszu kalkulacyjnym MS Excel. Zgodnie z założeniami, narzędzie to powinno pozwalać na szybkie uzyskanie raportu o układzie przedstawionym powyżej. Raport taki powinien zostać wygenerowany po skopiowaniu przez pracownika posiadanych danych źródłowych w określone miejsce arkusza będącego elementem proponowanego narzędzia. W niektórych przypadkach w celu wygenerowania raportu może zachodzić konieczność uzupełnienia przez niego brakujących danych. Ogólny schemat działania narzędzia został przedstawiony na rys. 31.3.

MPK	Koszt bez odliczenia VAT		Koszty z odliczeniem VAT		
			Użytek mieszany		Użytek firmowy
	8%	23%	8%	23%	

Rys. 31.2 Układ raportu zawierającego przetworzoną informację

Źródło: opracowanie własne



Rys. 31.3 Ogólny schemat proponowanego narzędzia

Źródło: opracowanie własne

31.3 STRUKTURA ENCJI

Na podstawie przeprowadzonej analizy, z uwzględnieniem założonej funkcjonalności narzędzia informatycznego, zaproponowano uwzględnienie w nim encji wymienionych i scharakteryzowanych w tab. 31.1.

Tab. 31.1 Ogólna charakterystyka encji zidentyfikowanych dla badanego problemu

Lp.	Nazwa encji	Opis
1	Pozycja	Pozycja w otrzymywanym od firmy zewnętrznej raporcie o kosztach użytkowania samochodów
2	Samochód	Samochód firmowy, w związku których generowane są koszty użytkowania
3	Produkt	Identyfikowany przez nazwę w otrzymywanym od firmy zewnętrznej raporcie, towar lub usługa, w związku z którą powstaje określonych koszt użytkowania samochodu
4	Grupa produktów	Grupa, do której zostaje przypisany każdy z produktów. Każda grupa ma jednoznacznie określoną stawkę podatku VAT oraz sposób jego wliczania do kosztu użytkowania samochodu.
5	MPK	Zgodne ze stosowanymi w firmie rozwiązaniami księgowymi, miejsce powstawania kosztów. W podziale na miejsca powstawania kosztów generowany jest raport wyjściowy z proponowanego narzędzia.


Źródło: opracowanie własne

Tab. 31.2 Zestawienie, ogólna charakterystyka oraz nazwy skrócone atrybutów zidentyfikowanych w ramach poszczególnych encji

Nazwa encji	Nazwa atrybutu	Charakterystyka
Pozycja	Rod karty	Pole znajdujące się w raporcie otrzymywanym od firmy zewnętrznej, w zakresie przeanalizowanych raportów przyjmowane wartości odnoszą się do nazw stacji paliw
	Numer faktury	Pole znajdujące się w raporcie otrzymywanym od firmy zewnętrznej, numer faktury, z której pochodzi transakcja prezentowana przez daną pozycję
	Nazwa klienta	Pole znajdujące się w raporcie otrzymywanym od firmy zewnętrznej, nazwa klienta, dla którego wystawiany jest raport w analizowanym przypadku atrybut ten zawsze przyjmuje jako wartość nazwę badanej firmy
	Data transakcji	Pole znajdujące się w raporcie otrzymywanym od firmy zewnętrznej, data transakcji prezentowanej przez daną pozycję
	Ilość	Pole znajdujące się w raporcie otrzymywanym od firmy zewnętrznej, ilość zakupionego towaru lub usługi
	Cena jednostkowa	Pole znajdujące się w raporcie otrzymywanym od firmy zewnętrznej, cena jednostkowa zakupionego towaru lub usługi
	Wartość netto	Pole znajdujące się w raporcie otrzymywanym od firmy zewnętrznej, wartość netto zakupionego towaru lub usługi
Samochód	Nr rejestracyjny	Numer rejestracyjny jednoznacznie identyfikujący samochód
	Użytek mieszany	Określa, czy samochód jest wykorzystywany wyłącznie do celów służbowych, czy zarówno do celów służbowych jak i firmowych. Wartość „0” oznacza wykorzystywanie samochodu jedynie do celów służbowych, wartość „1” oznacza użytek mieszany
Produkt	Nazwa produktu	Opis towaru lub usługi. Opis ten jednoznacznie identyfikuje produkt
Grupa	Nazwa grupy	Nazwa grupy produktów. Nazwa jednoznacznie identyfikuje grupę
	Odliczenie VAT	Atrybut określający, czy w przypadku produktów należących do danej grupy, podatek VAT jest odliczany, czy zaliczany do kosztów. Wartość „0” oznacza, że podatek VAT nie może zostać odliczony, wartość „1” oznacza, że podatek VAT może zostać odliczony
	Stawka VAT	Stawka podatku VAT, dla danej grupy produktowej. Atrybut może przyjmować wartość 8% albo 23%
MPK	Kod MPK	Zgodny ze stosowanym w firmie systemem księgowym kod miejsca powstawania kosztów

Źródło: opracowanie własne

Dla poszczególnych encji zidentyfikowane zostały określone atrybuty. Ich zestawienie wraz z krótką charakterystyką w tab. 31.2. Powiązania pomiędzy przedstawionymi encjami zostały przedstawione za pomocą modelu ERD na rys. 31.4. Model został stworzony zgodnie z metodologią CASE*Method [1, 9].



Rys. 31.4 Model związków encji dla zaproponowanego narzędzia informatycznego

Źródło: opracowanie własne

31.4 REPREZENTACJA ENCJI W ARKUSZU KALKULACYJNYM

Narzędzie informatyczne wspomagające proces rozliczania kosztów użytkowania samochodów zostało stworzone z wykorzystaniem najbardziej popularnego arkusza kalkulacyjnego, jakim jest MS Excel. Przyjęto, że zaproponowane rozwiązania powinny prawidłowo działać w programie od wersji MS Excel 2007. Poszczególne encje zostały przedstawione za pomocą tabel tak, jak ma to miejsce w klasycznej relacyjnej bazie danych. W związku z tym w odpowiednich kolumnach przedstawiane są wartości odpowiednich atrybutów, zaś w kolejnych wierszach pojawiają się kolejne wystąpienia danej encji. Każda tabela została umieszczona w oddzielnym arkuszu. Nazwy poszczególnych arkuszy odpowiadają nazwom reprezentowanych w nich encji. Poza podstawowymi atrybutami wymienionymi i scharakteryzowanymi w tab. 31.2, w przypadku niektórych tabel dodano kolumny, w których przechowywane są wartości atrybutów-kluczy obcych odzwierciedlających powiązania z innymi tabelami. Dodane atrybuty-klucze zostały przedstawione łącznie z ich krótką charakterystyką w tab. 31.3.

Tab. 31.3 Zestawienie oraz krótka charakterystyka atrybutów – kluczy obcych

Tabela	Związek	Kolumna	Opis
Pozycja	Pozycja-Samochód	Numer rejestracyjny	Wartość znajdująca się w raporcie otrzymywanym od firmy zewnętrznej, numer rejestracyjny samochodu, którego dotyczy transakcja prezentowana przez daną pozycję
	Pozycja-Produkt	Nazwa produktu	Wartość znajdująca się w raporcie otrzymywanym od firmy zewnętrznej, opis towaru lub usługi, której dotyczy prezentowana przez daną pozycję transakcja
Samochód	Samochód-MPK	Kod MPK	Kod miejsca powstawania kosztów, do którego przypisany jest dany samochód
Produkt	Produkt-Grupa	Nazwa grupy	Nazwa grupy, do której przypisany jest dany produkt

Źródło: opracowanie własne

Wszystkie stworzone w narzędziu tabele wraz z fragmentarycznymi fikcyjnymi danymi zostały przedstawione na rys. 31.5.

Arkusz "Pozycje"

	A	B	C	D	E	F	G	H	I	J
1	Rod Karty	Numer faktury	Numer karty	Numer rejestracyjny	Nazwa klienta	Data transakcji	Produkt	Ilość	Cena jednostkowa	Wartość netto
2	BP	194445624	487	DW9876E	Firma X	2015-05-11	Toll Service PL	1	14,00	14,00
3	BP	194445624	487	DW9876E	Firma X	2015-05-14	SupraDiesel LTR	52	3,90	202,80

Arkusz "Samochody"

	A	B	C
1	Numer rejestracyjny	MPK	Użytek mieszany
2	DW9876E	1500430740	1
3	DW7534C	1500430750	0

Arkusz "Grupy"

	A	B	C
1	Nazwa	Odliczenie VAT	Stawka VAT
2	Toll Service	1	0,23
3	Paliwo	0	0,23

Arkusz "Produkty"

	A	B
1	Nazwa produktu	Nazwa grupy
2	Toll Service PL	Toll Service
3	SupraDiesel LTR	Paliwo

Arkusz "MPK"

	A
1	Kod MPK
2	1500430740
3	1500430750

Rys. 31.5 Układ tabel reprezentujących poszczególne encje w arkuszu kalkulacyjnym

Źródło: opracowanie własne

W omawianym narzędziu założono, że tabela „Pozycje” może maksymalnie zawierać dane do wiersza nr 1000, tabela „Samochody” do wiersza nr 100, tabela „Produkty” do wiersza nr 500, tabela „Grupy” do wiersza nr 500 i tabela MPK do wiersza nr 20. Do tak założonych zakresów wierszy odnosić się będą przedstawiane w dalszej części niniejszego opracowania formuły.

31.5 INTEGRALNOŚĆ DANYCH W POSZCZEGÓLNYCH TABELACH

Jednym z bardzo ważnych problemów w proponowanym narzędziu jest zapewnienie pełnej integralności danych wprowadzanych we wszystkich tabelach. Dla prawidłowego funkcjonowania omawianego rozwiązania, wartości atrybutów będących kluczami obcymi muszą być w pełni zgodne z wartościami odpowiednich kluczy podstawowych z tabel źródłowych. Rozpatrując problem integralności danych w tabelach można wyróżnić dwa przypadki szczegółowe. Pierwszy z nich dotyczy sytuacji, w której wartości wprowadzane, jako klucze obce muszą pochodzić z istniejącego już zbioru wartości wprowadzonych wcześniej w odpowiedniej tabeli jako klucze podstawowe. Taka sytuacja występuje przy wprowadzaniu wartości atrybutu „MPK” w tabeli „Samochody” oraz atrybutu „Nazwa grupy” w tabeli „Produkty”. Drugi przypadek dotyczy sytuacji, w której do tabeli, jako klucze obce, mogą zostać wprowadzone dowolne wartości, jednak w przypadku wprowadzenia wartości, które nie występują jeszcze w odpowiednich tabelach, jako klucze podstawowe, klucze te, w tabelach źródłowych muszą zostać uzupełnione. Taka sytuacja występuje podczas wprowadzania atrybutów „Numer rejestracyjny” i „Produkt” w tabeli „Pozycje”.

W pierwszym przypadku, w celu zapewnienia spójności danych wykorzystano wbudowany w arkuszu kalkulacyjnym mechanizm „Sprawdzania poprawności danych” w trybie ograniczania zakresu wartości możliwych do wprowadzenia do komórki. Dzięki

zastosowaniu wspomnianego mechanizmu użytkownik może wprowadzać wartości atrybutów w odpowiednich komórkach używając rozwijanych list z wartościami dopuszczalnymi. W celu zdefiniowania zakresów źródłowych wykorzystano mechanizm „Mendżera nazw”. Określając odpowiednie nazwane obszary arkusza wykorzystano funkcje „PRZESUNIĘCIE()” i „ILE.NIEPUSTYCH()” tak, żeby obejmowały one jedynie te obszary, do których wprowadzono wcześniej odpowiednie wartości. Dzięki temu na pojawiających się rozwijanych listach pomijane są puste komórki, a mechanizm „Sprawdzania poprawności danych” rzeczywiście ogranicza zakres możliwych do wprowadzenia wartości. Użyte w narzędziu, w ramach pierwszego omawianego przypadku zapewnienia integralności danych, nazwy obszarów łącznie z ich definicjami oraz nazwami tabel i kolumn, w których zostały zastosowane w ramach mechanizmu „Sprawdzania poprawności danych”, przedstawiono w tab. 31.4.

Tab. 31.4 Nazwane obszary arkusza wykorzystane w pierwszym omawianym przypadku zapewnienia integralności danych w tabelach

Obszar	Odwołanie	Tabela, kolumna
Grupa	=PRZESUNIĘCIE(Grupy!\$A\$2;0;0;ILE.NIEPUSTYCH(Grupy!\$A\$2:\$A\$50);1)	Produkty, nazwa grupy
MPK	=PRZESUNIĘCIE(MPK!\$A\$2;0;0;ILE.NIEPUSTYCH(MPK!\$A\$2:\$A\$20);1)	Samochody, MPK

Źródło: opracowanie własne

W drugim wspomnianym przypadku, wartości kluczy obcych w tabeli „Pozycje”, podobnie jak i wszystkie inne wartości w tej tabeli, są wprowadzane poprzez skopiowanie danych z raportu uzyskanego od firmy zewnętrznej. Występowanie w tym raporcie wartości atrybutów „Numer rejestracyjny” oraz „Produkt”, które jeszcze nie istnieją w tabelach „Samochody” oraz „Produkty” jest bardzo prawdopodobne. Ograniczanie zakresu możliwych do wprowadzenia wartości nie byłoby w tym przypadku dobrym rozwiązaniem. Zamiast tego istotne jest, żeby po skopiowaniu wszystkich wartości użytkownik był w stanie szybko zidentyfikować fakt, że wśród kluczy obcych występują wartości, które nie mają swoich odpowiedników w zbiorach kluczy podstawowych w tabelach „Samochody” i „Produkty”, co jest związane z koniecznością uzupełnienia wpisów w tych tabelach. W celu zapewnienia tej funkcjonalności zaproponowano wykorzystanie mechanizmu „Formatowania warunkowego”.

	A	B	C	D	G	Q	R
1	Rod Karty	Numer faktury	Numer karty	Numer rejestracyjny	Produkt	Czy jest numer rejestracyjny	Czy jest produkt
2	BP	194445624	487	DW9876E	Toll Service PL	1	1
3	BP	194445624	487	DW9876E	SupraDiesel LTR	1	1
25	BP	194445664	481	DW99132K	MYJNIA ZETON	0	0
26	BP	194445664	481	DW99132K	Toll Service PL	0	1
FA	=LICZ.JEŻELI(Samochody!\$A\$2:\$A\$100;Pozycje!D2)						
FB	=LICZ.JEŻELI(Produkty!\$A\$2:\$A\$500;Pozycje!G2)						

Rys. 31.6 Zastosowanie mechanizmu „Formatowania warunkowego” do identyfikacji kluczy obcych wymagających uzupełnienia

Źródło: opracowanie własne

Do określenia komórek, które powinny zostać za pomocą tego mechanizmu wyróżnione, w arkuszu „Pozycje” dodano dwie kolumny pomocnicze o nazwach „Czy jest numer rejestracyjny” oraz „Czy jest produkt”. Kolumny pomocnicze wraz z zastosowanymi formułami zostały przedstawione na rys. 31.6. Na potrzeby prezentacji tych oraz wszystkich pozostałych formuł w niniejszym opracowaniu przyjęto zasadę, że są one prezentowane w postaci, jaką przyjmują w lewej górnej komórce obszaru, w którym są wprowadzone.

Pojawiające się w dodanych kolumnach wartości „0” oznaczają, że w danym wierszu nie zostały wprowadzone żadne wartości dla numerów rejestracyjnych lub produktów albo, że wprowadzone wartości nie występują odpowiednio w tabelach „Samochody” lub „Produkty”. W drugim przypadku komórki z wprowadzonymi numerami rejestracyjnymi samochodów lub nazwami produktów powinny zostać wyróżnione. Zastosowane w tym celu reguły formatowania warunkowego zostały przedstawione na rys. 31.7.

Reguła (stosowana w następującej kolejności)	Formatowanie	Dotyczy
Formuła: =ORAZ(R2=0;G2<>"")	AaBbCcYyZz	=\$G\$2:\$G\$1000
Formuła: =ORAZ(Q2=0;D2<>"")	AaBbCcYyZz	=\$D\$2:\$D\$1000

Rys. 31.7 Reguły mechanizmu „Formatowania warunkowego” zastosowane w arkuszu „Pozycje”

Źródło: opracowanie własne

Dodatkowo, oprócz zasygnalizowania konieczności wprowadzenia brakujących wartości w tabelach „Pozycje” lub „Samochody”, w narzędziu zaproponowano rozwiązanie ułatwiające sam proces wprowadzania. Na potrzeby tego rozwiązania w narzędziu dodany został dodatkowy arkusz „T”, w którym na bieżąco generowane są zestawienia numerów rejestracyjnych oraz nazw produktów występujących w tabeli „Pozycje” oraz niewystępujących w tabelach „Samochody” lub „Produkty”. Rozbudowany został także arkusz „Pozycje”, w którym dodane zostały cztery kolumny o nazwach „Brakujące numery rejestracyjne T1”, „Brakujące numery rejestracyjne T2”, „Brakujące produkty T1” oraz „Brakujące produkty T2”. Dodatkowe kolumny w arkuszu „Pozycje” wraz z zastosowanymi w nich formułami zostały przedstawione na rys. 31.8.

Układ kolumn, wraz z zastosowanymi w nich formułami, w dodatkowo stworzonym arkuszu „T” został przedstawiony na rys. 31.9. W arkuszu „Pozycje”, w kolumnie „Brakujące numery rejestracyjne T1”, formuła „FC” zwraca numer wystąpienia w tabeli „Pozycje”, licząc od początku tabeli, danego numeru rejestracyjnego. W kolumnie „Brakujące numery rejestracyjne T2” formuła „FD” zwraca kolejną liczbę naturalną przy każdym numerze rejestracyjnym niewystępującym w tabeli „Samochody” i pojawiającym się w tabeli „Pozycje” po raz pierwszy. W przypadku, gdy dla wartości w bieżącym wierszu którykolwiek z wymienionych warunków nie jest spełniony formuła zwraca liczbę taką samą, jak w wierszu poprzednim. W analogiczny sposób formuły „FE” i „FF” zwracają odpowiednie liczby dla nazw produktów. W arkuszu „T”, w kolumnie Lp. wprowadzone zostały kolejne liczby naturalne. Wprowadzona w kolumnie „Brakujące numery

rejestracyjne” formuła „FG”, w kolejnych wierszach zwraca kolejne numery rejestracyjne, przy których w arkuszu „Pozycje”, w kolumnie „Brakujące numery rejestracyjne T2” pojawiły się kolejne liczby naturalne. W ten sposób tworzona jest lista numerów rejestracyjnych, które należy wprowadzić do tabeli „Samochody”.

	D	G	Q	R	T	U	V	W
	Numer rejestracyjny	Produkt	Czy jest numer rejestracyjny	Czy jest produkt	Brakujące numery rejestracyjne	Brakujące numery rejestracyjne	Brakujące produkty	Brakujące produkty
1					FC T1	FD T2	FE T1	FF T2
2	DW9876E	Toll Service PL	1	1	1	0	1	0
3	DW9876E	SupraDiesel LTR	1	1	2	0	1	0
4	DW9876E	MYJNIA ZETON	1	0	3	0	1	1
5	DW9876E	PLYN LETNI NEW	1	0	4	0	1	2
6	DW9876E	Toll Service P	1	1	5	0	1	2
FC	=LICZ.JEŻELI(\$D\$2:D2;D2)							
FD	=LICZ.WARUNKI(\$Q\$2:Q2;0;\$T\$2:T2;1)							
FE	=LICZ.JEŻELI(\$G\$2:G2;G2)							
FF	=LICZ.WARUNKI(\$R\$2:R2;0;\$V\$2:V2;1)							

Rys. 31.8 Dodatkowe kolumny i formuły rozwiązania ułatwiającego wprowadzanie brakujących wartości

Źródło: opracowanie własne

	A	B	C
	L.p.	Brakujące numery rejestracyjne	Brakujące produkty
1		FG	FH
2	1	DW99132K	MYJNIA ZETON
3	2		PLYN LETNI NEW
4	3		KOSZTY VAS
5	4		
FG	=JEŻELI(LICZ.JEŻELI(Pozycje!\$U\$2:\$U\$1000;T!A2)>0; INDEKS(Pozycje!\$D\$2:\$D\$1000;PODAJ.POZYCJE(T!A2;Pozycje!\$U\$2:\$U\$1000;0);1; ""))		
FH	=JEŻELI(LICZ.JEŻELI(Pozycje!\$W\$2:\$W\$1000;T!A2)>0; INDEKS(Pozycje!\$G\$2:\$G\$1000;PODAJ.POZYCJE(T!A2;Pozycje!\$W\$2:\$W\$1000;0);1; ""))		

Rys. 31.9 Układ kolumn oraz formuły w arkuszu „T”

Źródło: opracowanie własne

W identyczny sposób, za pomocą formuły „FH” tworzona jest lista brakujących nazw produktów. Oba utworzone zestawienia wykorzystywane są, jako listy źródłowe dla zastosowanego również w tym przypadku mechanizmu „Sprawdzania poprawności danych”. Do wspomaganego wprowadzania brakujących wartości atrybutu „Numer rejestracyjny” w tabeli „Samochody” oraz atrybutu „Nazwa produktu” w tabeli „Produkty”, mechanizm ten został zastosowany w trybie podpowiadania wartości możliwych do wprowadzenia. Mogą one, więc być tutaj wprowadzone zarówno z wykorzystaniem roz-

wijanej listy jak i wpisywane w sposób ręczny bez żadnych ograniczeń. Także w tym przypadku zakresy źródłowe dla rozwijanych list zostały określone z wykorzystaniem „Menedżera nazw”. Do zdefiniowania odpowiednich nazwanych obszarów arkusza wykorzystane zostały funkcje PRZESUNIĘCIE() i MAX(). Nazwy obszarów zastosowane w tym przypadku wraz z ich definicjami oraz nazwami tabel i kolumn, w których zostały zastosowane w ramach mechanizmu „Sprawdzania poprawności danych”, przedstawiono w tab. 31.5.

Tab. 31.5 Nazwane obszary arkusza wykorzystane w drugim omawianym przypadku zapewnienia integralności danych w tabelach

Obszar	Odwołanie	Tabela i kolumna
BrakNumRej	=PRZESUNIĘCIE(T!\$B\$2;0;0;MAX(Pozycje!\$U\$2:\$U\$1000);1)	Samochody, nr rejestracyjny
BrakProd	=PRZESUNIĘCIE(T!\$C\$2;0;0;MAX(Pozycje!\$W\$2:\$W\$1000);1)	Produkty, nazwa produktu

Źródło: opracowanie własne

31.6 GENEROWANIE RAPORTU

Po zapewnieniu spójności zgromadzonych w narzędziu danych, możliwe jest wygenerowanie założonego raportu zgodnego z układem przedstawionym na rys. 31.2. Na potrzeby stworzenia tego raportu, w arkuszach „MPK” oraz „Pozycje” dodane zostały dodatkowe kolumny pomocnicze, które wraz z zastosowanymi formułami zostały przedstawione na rys. 31.10 i 31.11.

	A	B	C	D
1	Kod MPK	FI	T1	FJ
2	1500430740		1	1
3	1500430750		1	2
4	1500430427		0	2
5	1500430425		1	3

FI	=JEŻELI(ORAZ(A2<>"";LICZ.JEŻELI(Pozycje!\$O\$2:\$O\$1000;MPK!A2)>0);1;0)
FJ	=SUMA(\$C\$2:C2)

Rys. 31.10 Kolumny pomocnicze oraz zestawienie formuł w arkuszu „MPK”

Źródło: opracowanie własne

	J	L	M	N	O	P
	Wartość netto	Grupa	Odliczenie VAT	Stawka VAT	MPK	Użytek mieszany
1		FK	FL	FM	FN	FO
2	14,00	Toll Service	1	0,23	1500430740	1
3	202,80	Paliwo	0	0,23	1500430740	1

FK	=WYSZUKAJ.PIONOWO(G2;Produkty!\$A\$2:\$B\$500;2;FAŁSZ)
FL	=WYSZUKAJ.PIONOWO(L2;Grupy!\$A\$2:\$C\$50;2;FAŁSZ)
FM	=WYSZUKAJ.PIONOWO(L2;Grupy!\$A\$2:\$C\$50;3;FAŁSZ)
FN	=WYSZUKAJ.PIONOWO(D2;Samochody!\$A\$2:\$B\$100;2;FAŁSZ)
FO	=WYSZUKAJ.PIONOWO(D2;Samochody!\$A\$2:\$C\$100;3;FAŁSZ)

Rys. 31.11 Kolumny pomocnicze oraz zestawienie formuł w arkuszu „Pozycje”

Źródło: opracowanie własne

Znajdujące się w arkuszu „MPK” formuły „FI” oraz „FJ” służą do zidentyfikowania numerów MPK, dla których w raporcie otrzymywanym od firmy zewnętrznej istnieją jakieś koszty i które mają zostać uwzględnione w generowanym raporcie końcowym. Pierwsza z formuł zwraca wartość „1” przy tych numerach MPK, których dotyczą dane wejściowe. Druga formuła, przy każdym takim kolejnym numerze MPK zwraca kolejną liczbę naturalną. W przypadku numeru MPK, który nie ma pojawić się w raporcie, powtarzana jest poprzednia liczba naturalna. Formuły przedstawione na rys. 31.11 zwracają dla każdej pozycji zbiór atrybutów niezbędnych do przypisania znajdującej się w niej wartości netto do odpowiedniego miejsca w raporcie końcowym.

Raport końcowy tworzony jest w arkuszu „Raport1”. Układ kolumn w tym arkuszu wraz z zastosowanymi formułami został przedstawiony na rys. 31.12.

W arkuszu tym obszarze „C4:G6” wprowadzone zostały wartości pomocnicze, wykorzystane w odpowiednich formułach. W ostatecznej postaci raportu wiersze 4:6 mogą zostać ukryte. Formuła „FP” przenosi do raportu te numery MPK, przy których w kolumnie „T2” w arkuszu „MPK” znajdują się pierwsze wystąpienia kolejnych liczb naturalnych. Formuły „FR”, „FS” i „FT” na podstawie numerów MPK oraz wartości pomocniczych ustalają wysokość kosztu obciążającego poszczególne miejsca powstawania kosztów.

	A	B	C	D	E	F	G
1			Koszt bez odliczenia VAT		Koszt z odliczeniem VAT		
2	Lp	MPK			Użytek mieszany		Użytek firmowy
3			8%	23%	8%	23%	
4	Wartości pomocnicze				1	1	0
5			0	0	1	1	1
6		FP	FR 0,08	0,23	FS 0,08	0,23	FT
7	1	1500430740	0,00	249,44	0,00	31,22	0,00
8	2	1500430750	0,00	249,44	0,00	0,00	27,11

FP	=JEŻELI(LICZ.JEŻELI(MPK!\$D\$2:\$D\$20;Raport1!A7)>0; INDEKS(MPK!\$A\$2:\$A\$20;PODAJ.POZYCJĘ(Raport1!A7;MPK!\$D\$2:\$D\$20;0);1); "")
FR	=JEŻELI(B7<>""); SUMA.WARUNKÓW(Pozycje!\$J\$2:\$J\$1000;Pozycje!\$M\$2:\$M\$1000;Raport1!CS5;Pozycje!\$O\$2:\$O\$1000; Raport1!\$B7;Pozycje!\$N\$2:\$N\$1000;Raport1!CS6)*(1+CS6); "")
FS	=JEŻELI(B7<>""); SUMA.WARUNKÓW(Pozycje!\$J\$2:\$J\$1000;Pozycje!\$M\$2:\$M\$1000;Raport1!E\$5;Pozycje!\$O\$2:\$O\$1000; Raport1!\$B7;Pozycje!\$N\$2:\$N\$1000;Raport1!E\$6;Pozycje!\$P\$2:\$P\$1000;Raport1!E\$4)*(1+0,5*E\$6); "")
FT	=JEŻELI(B7<>""); SUMA.WARUNKÓW(Pozycje!\$J\$2:\$J\$1000;Pozycje!\$M\$2:\$M\$1000;Raport1!G\$5;Pozycje!\$O\$2:\$O\$1000; Raport1!\$B7;Pozycje!\$P\$2:\$P\$1000;Raport1!G\$4); "")

Rys. 31.12 Układ kolumn oraz formuły w generowanym raporcie końcowym

Źródło: opracowanie własne

PODSUMOWANIE

Zaprezentowane rozwiązanie pokazuje, że arkusz kalkulacyjny traktowany jako środowisko, jest w zupełności wystarczający do stworzenia narzędzia, które ma wspomagać omówiony na wstępie proces rozliczania kosztów samochodów na podstawie

raportu otrzymywanego od firmy zewnętrznej. Za jego pomocą można stworzyć rozwiązanie bazujące na relacyjnym modelu danych, którego reprezentacja w arkuszu kalkulacyjnym może się odbywać za pomocą tabel takich samych jak ma to miejsce w klasycznej relacyjnej bazie danych. Dostępne w arkuszu kalkulacyjnym mechanizmy takie jak mechanizm „Sprawdzania poprawności danych” czy mechanizm „Formatowania warunkowego” pozwalają na zapewnienie integralności danych przechowywanych w poszczególnych tabelach. Założona funkcjonalność narzędzia była możliwa do uzyskania przy wykorzystaniu bardzo ograniczonego zbioru elementów arkusza kalkulacyjnego i w żadnym miejscu nie wymagała użycia kodu napisanego w języku programowania VBA. Zastosowane elementy arkusza obejmują mechanizm „Sprawdzania poprawności danych”, „Menedżera nazw”, mechanizm „Formatowania warunkowego” oraz funkcje takie jak: PRZESUNIĘCIE(), ILE.NIEPUSTYCH(), LICZ.JEŻELI(), ORAZ(), LICZ.WARUNKI(), JEŻELI(), LUB(), INDEKS(), PODAJ.POZYCJĘ(), MAX(), SUMA(), WYSZUKAJ.PIONOWO(). Nie wielki zakres wymienionych elementów, pozwala sądzić, że narzędzia podobnego typu mogłyby być bez większych problemów opracowywane samodzielnie przez pracowników uczestniczących w procesach, w ramach których konieczne jest gromadzenie i przetwarzanie informacji a które z powodu braku odpowiednich rozwiązań informatycznych realizowane są ręcznie.

LITERATURA

- 1 R. Berker. *CASE Method, Modelowanie związków encji*. Warszawa: WNT, 1996.
- 2 C. Carlberg. *Microsoft Excel 2007 PL. Analizy biznesowe. Rozwiązania w biznesie*. Gliwice: Helion. 2009.
- 3 M. Jackson, M. Staunton. *Zaawansowane modele finansowe z wykorzystaniem Excela i VBA*. Gliwice: Helion, 2004.
- 4 Z. Klonowski. *Systemy informatyczne zarządzania przedsiębiorstwem, modele rozwoju i właściwości funkcjonalne*, Wrocław: Oficyna Wydawnicza Politechniki Wrocławskiej, 2004
- 5 G. Knight. *Excel. Analiza danych biznesowych*. Gliwice: Helion, 2006.
- 6 K. Masłowski. *Excel. Funkcje w przykładach*. Gliwice: Helion, 2007.
- 7 J. Niezgodziński. „Analiza wrażliwości modelu finansowego inwestycji z wykorzystaniem programu Excel.” *Controlling i Rachunkowość Zarządcza*, nr 1, 2000. s. 15-19.
- 8 D.J. Power. „A Brief History of Spreadsheets”. Pobrane z: <http://www.dssresources.com/history/sshistory.html> [Dostęp: 10.09.2015].
- 9 A. Sharon. *Modelowanie danych*. Gliwice: Helion, 2006.
- 10 B. Szczeńiak. „Arkusz kalkulacyjny w doskonaleniu procesu układania planu zajęć w szkole specjalnej”, R. Knosala (red.) *Komputerowo zintegrowane zarządzanie, t.2*. Opole: Oficyna Wydawnicza Polskiego Towarzystwa Zarządzania Produkcją, 2010.
- 11 B. Szczeńiak, A. Bujanowska. „Koncepcja zastosowania arkusza kalkulacyjnego do wspomaganie procesu przeglądów urządzeń w wybranym szpitalu.” *Studia i Materiały Polskiego Stowarzyszenia Zarządzania Wiedzą*, t. 45, Bydgoszcz: PSZW, 2011.

- 12 B. Szczęśniak, A. Bujanowska. „Koncepcja zastosowania arkusza kalkulacyjnego do wspomaganie realizacji procesu diagnostyki oraz napraw w wybranym szpitalu.” *Systemy Wspomagania w Inżynierii Produkcji*, nr 2(2), 2012.
- 13 B. Szczęśniak. „Concept of supportive spreadsheet application in the survey of production departments’ satisfaction with services of maintenance departments.” *Scientific Journals Maritime University of Szczecin*, No 32(104), 2012.
- 14 B. Szczęśniak. „Koncepcja zastosowania arkusza kalkulacyjnego do wspomaganie tworzenia dokumentów w procesie produkcji taśm blachy.” *Systemy Wspomagania w Inżynierii Produkcji. Innowacyjność, Jakość, Zarządzanie*, nr 2(4), 2013.
- 15 B. Szczęśniak. „Koncepcja zastosowania arkusza kalkulacyjnego do wspomaganie prowadzenia kart kontrolnych Shewarta”. *Systemy Wspomagania w Inżynierii Produkcji. Jakość i Bezpieczeństwo*, nr 2(8), 2014, s. 208-225.
- 16 B. Szczęśniak, K. Michalski. „Spreadsheet as a means to support a timetable scheduling process with reference to the example of a course entitled with reference to the example of a course entitled "education of specialists in the field of management of post-mining areas in the Polish-Czech borderland". *15th International Multidisciplinary Scientific GeoConference SGEM 2015. Ecology, economics, education and legislation*, Conference proceedings: STEF92 Technology, 2015, p. 315-322.
- 17 B. Szczęśniak, M. Molenda. „Spreadsheet application supporting the x-r control chart.” *Conference Proceedings - 22th Conference Modern Mathematical Methods in Engineering (3mi)*, June 3 - 5, 2013 Horní Lomná, Czech Republic.
- 18 B. Szczęśniak. „Zastosowanie arkusza kalkulacyjnego do wspomaganie metody ABC”. *Zeszyty Naukowe Politechniki Śląskiej*, seria Organizacja i Zarządzanie z.50, 2010.
- 19 J. Tyszkiewicz. „Spreadsheet As a Relational Database Engine”, *Proceedings of the 2010 ACM SIGMOD International Conference on Management of data*, Indianapolis, IN, USA — June 06 - 11, 2010.
- 20 J. Walkenbach. *Excel 2010 PL. Biblia*. Gliwice: Helion, 2011.
- 21 M. Wichary. „Historia arkuszy kalkulacyjnych – Ćwierć wieku za kratkami”. *Chip*, nr 10, 2004.
- 22 S. Wilczewski, M. Wrzód. *Excel 2007 w firmie. Controlling, finanse i nie tylko*. Gliwice: Helion. 2008.
- 23 W.L. Winston. *Microsoft Excel 2010. Data Analysis and Business Modeling*. Washington: Microsoft Press, 2011.

WYKORZYSTANIE ARKUSZA KALKULACYJNEGO DO WSPOMAGANIA PROCESU ROZLICZANIA KOSZTÓW UŻYTKOWANIA SAMOCHODÓW

Streszczenie: W artykule zaprezentowano bazujące na relacyjnym modelu danych narzędzie informatyczne wspomagające proces rozliczania kosztów użytkowania samochodów. W pierwszej kolejności omówiono proces rozliczania kosztów. Przedstawiono strukturę danych wejściowych, pożądany układ raportu końcowego oraz reguły, które podczas tworzenia takiego raportu muszą być uwzględnione. Dla przeanalizowanego problemu zidentyfikowano oraz opisano strukturę encji. W dalszej kolejności opisano sposób reprezentacji encji w arkuszu kalkulacyjnym oraz omówiono zastosowane rozwiązania zapewniające integralność danych w przechowywanych w narzędziu. Na koniec przedstawiono rozwiązania pozwalające wygenerować w sposób automatyczny końcowe zestawienie kosztów. Założoną funkcjonalność narzędzia uzyskano przy wykorzystaniu dość wąskiego zakresu funkcji i mechanizmów arkusza kalkulacyjnego MS Excel, bez konieczności użycia kodu napisanego w języku programowania VBA.

Słowa kluczowe: doskonalenie procesów, procesy informacyjne, relacyjny model danych, arkusz kalkulacyjny

APPLICATION OF A SPREADSHEET TO SUPPORT THE PROCESS OF CAR USE COSTS ACCOUNTING

Abstract: This article presents an IT tool based on the relational data model, developed in order to support the process of car use costs accounting. Firstly, the manner in which the costs in question are accounted has been explained. The input data structure has been discussed as well as the intended final report layout and the principles which must be taken into consideration while such a report is being created. With regard to the problem analysed, the structure of entities has been identified and described. Further on in the article, the method used to represent entities in a spreadsheet has been described and the solutions ensuring integrity of the data stored in the tool have been elaborated upon. In the closing paragraphs of the paper, some more solutions enabling automatic generation of the final cost breakdown have been introduced. The assumed functionality of the tool in question has been attained by means of a relatively narrow array of functions and mechanisms featured by the MS Excel spreadsheet, without the need for application of a code written in the VBA programming language.

Key words: process support, information processes, relational data model, spreadsheet

Dr inż. Bartosz SZCZĘŚNIAK
Politechnika Śląska
Wydział Organizacji i Zarządzania
Instytut Inżynierii Produkcji
ul. Roosevelta 26, 41-800 Zabrze
e-mail: Bartosz.Szczęśniak@polsl.pl

Data przesłania artykułu do Redakcji: 15.04.2016
Data akceptacji artykułu przez Redakcję: 13.05.2016